

A Two-Body Problem with Gravity Forces

```
> restart;
> with(DEtools):
> with(LinearAlgebra):
> with(plots):
```

Start: Two Bodies Connected by a Spring (without mass)

`m1` and `m2` define the masses of the two bodies:

```
> m1 := 1; m2 := 2;

                                m1 := 1
                                m2 := 2                                (1.1)
```

The connecting spring applies a force in the direction of the connecting vector $\mathbf{q}-\mathbf{r}$ (\mathbf{q} and \mathbf{r} the position vectors of the bodies):

```
> spring_ode := { m1*diff( q[1](t),t,t ) = r[1](t)-q[1](t),
                  m1*diff( q[2](t),t,t ) = r[2](t)-q[2](t),
                  m2*diff( r[1](t),t,t ) = q[1](t)-r[1](t),
                  m2*diff( r[2](t),t,t ) = q[2](t)-r[2](t)};
```

$$\text{spring_ode} := \left\{ 2 \left(\frac{d^2}{dt^2} r_1(t) \right) = q_1(t) - r_1(t), 2 \left(\frac{d^2}{dt^2} r_2(t) \right) = q_2(t) - r_2(t), \frac{d^2}{dt^2} q_1(t) \right. \quad (1.2)$$

$$\left. = r_1(t) - q_1(t), \frac{d^2}{dt^2} q_2(t) = r_2(t) - q_2(t) \right\}$$

Note that the initial positions of the bodies alone are not sufficient as initial conditions to compute the solution:

```
> dsolve( { op(spring_ode), q[1](0)=1, q[2](0)=0, r[1](0)=0, r
           [2](0)=1},
           {q[1](t), q[2](t), r[1](t), r[2](t)
           });
```

$$\left\{ q_1(t) = \frac{1}{3} + _C6 t + _C7 \sin\left(\frac{1}{2} \sqrt{6} t\right) + \frac{2}{3} \cos\left(\frac{1}{2} \sqrt{6} t\right), q_2(t) = \frac{2}{3} + _C2 t \right. \quad (1.3)$$

$$\left. + _C3 \sin\left(\frac{1}{2} \sqrt{6} t\right) - \frac{2}{3} \cos\left(\frac{1}{2} \sqrt{6} t\right), r_1(t) = -\frac{1}{2} _C7 \sin\left(\frac{1}{2} \sqrt{6} t\right) \right.$$

$$\left. - \frac{1}{3} \cos\left(\frac{1}{2} \sqrt{6} t\right) + \frac{1}{3} + _C6 t, r_2(t) = -\frac{1}{2} _C3 \sin\left(\frac{1}{2} \sqrt{6} t\right) \right.$$

$$\left. + \frac{1}{3} \cos\left(\frac{1}{2} \sqrt{6} t\right) + \frac{2}{3} + _C2 t \right\}$$

Hence, we need to provide the body velocities (first derivatives) as additional initial conditions:

```
> qrsol := dsolve( { op(spring_ode),
                   q[1](0)=1, q[2](0)=0, D(q[1])(0) = 0, D(q[2])(0) =
                   1,
                   r[1](0)=0, r[2](0)=2, D(r[1])(0) = 0.5, D(r[2])(0)
                   = 0 },
```

```

      {q[1](t), q[2](t), r[1](t), r[2](t)}):
> qs1 := subs(qrsol, q[1](t)); qs2 := subs(qrsol, q[2](t));
  rs1 := subs(qrsol, r[1](t)); rs2 := subs(qrsol, r[2](t));
  qs1 :=  $\frac{1}{3} + \frac{1}{3}t - \frac{1}{9}\sin\left(\frac{1}{2}\sqrt{6}t\right)\sqrt{6} + \frac{2}{3}\cos\left(\frac{1}{2}\sqrt{6}t\right)$ 
  qs2 :=  $\frac{4}{3} + \frac{1}{3}t + \frac{2}{9}\sin\left(\frac{1}{2}\sqrt{6}t\right)\sqrt{6} - \frac{4}{3}\cos\left(\frac{1}{2}\sqrt{6}t\right)$ 
  rs1 :=  $\frac{1}{18}\sin\left(\frac{1}{2}\sqrt{6}t\right)\sqrt{6} - \frac{1}{3}\cos\left(\frac{1}{2}\sqrt{6}t\right) + \frac{1}{3} + \frac{1}{3}t$ 
  rs2 :=  $-\frac{1}{9}\sin\left(\frac{1}{2}\sqrt{6}t\right)\sqrt{6} + \frac{2}{3}\cos\left(\frac{1}{2}\sqrt{6}t\right) + \frac{4}{3} + \frac{1}{3}t$ 


```

(1.4)

```

> qplot := plot( [ qs1,qs2, t=0..10], scaling=CONSTRAINED,
  colour=blue, thickness=2 ):
  rplot := plot( [ rs1,rs2, t=0..10], scaling=CONSTRAINED,
  color=red, thickness=2 ):
  display(qplot,rplot);

```



Computing a Numerical Solution (using Maple)

Define the initial conditions:

(note: to make sure that the entire system does not move, the total momentum has to be 0)

```
> grincon :=
```

```
  [[q[1](0)=1, q[2](0)=0, D(q[1])(0) = 0, D(q[2])(0) = 1,
```


```
    r[1](0)=0, r[2](0)=2, D(r[1])(0) = 0, D(r[2])(0) = -0.5]
```

```
  ];
```

```
grincon := [[q1(0) = 1, q2(0) = 0, D(q1)(0) = 0, D(q2)(0) = 1, r1(0) = 0, r2(0) = 2, D(r1)(0) = 0, D(r2)(0) = -0.5] (1.1.1)
```

```
=2, D(r1)(0) = 0, D(r2)(0) = -0.5]]
```

```
> Tend := 4:  
qnum := DEplot(spring_ode, [q[1],q[2],r[1],r[2]], t=0..  
Tend, qincon,  
                                  stepsize=0.2, scene=[q[1],q[2]], linecolor=  
blue):  
rnum := DEplot(spring_ode, [q[1],q[2],r[1],r[2]], t=0..  
Tend, qincon,  
                                  stepsize=0.2, scene=[r[1],r[2]], linecolor=  
red):  
display(qnum,rnum);
```



Two-Body Problem with Gravity Forces

```
> m1 := 1; m2 := 6;
```

```
      m1 := 1
```

```
      m2 := 6
```

(2.1)

Define two functions to compute the gravity forces (q_1 and q_2 contain the coordinates of body #1; similar for r_1 and r_2) -
the gravity forces will always be directed along the connection vector:

```
> Fg1 := (q1,q2,r1,r2) -> -m1*m2/((q1-r1)^2 + (q2-r2)^2 )^
(3/2) *(q1-r1);
Fg2 := (q1,q2,r1,r2) -> -m1*m2/((q1-r1)^2 + (q2-r2)^2 )^
(3/2) *(q2-r2);
```

$$Fg1 := (q1, q2, r1, r2) \rightarrow -\frac{m1 m2 (q1 - r1)}{((q1 - r1)^2 + (q2 - r2)^2)^{3/2}}$$

$$Fg2 := (q1, q2, r1, r2) \rightarrow -\frac{m1 m2 (q2 - r2)}{((q1 - r1)^2 + (q2 - r2)^2)^{3/2}} \quad (2.2)$$

Set up the system of ODE for the gravity problem:

```
> grav_ode := { m1*diff( q[1](t),t,t ) = Fg1(q[1](t),q[2](t),
r[1](t),r[2](t)),
                m1*diff( q[2](t),t,t ) = Fg2(q[1](t),q[2](t),
r[1](t),r[2](t)),
                m2*diff( r[1](t),t,t ) = Fg1(r[1](t),r[2](t),
q[1](t),q[2](t)),
                m2*diff( r[2](t),t,t ) = Fg2(r[1](t),r[2](t),
q[1](t),q[2](t)) };
```

$$grav_ode := \left\{ \begin{aligned} 6 \left(\frac{d^2}{dt^2} r_1(t) \right) &= -\frac{6 (r_1(t) - q_1(t))}{((r_1(t) - q_1(t))^2 + (r_2(t) - q_2(t))^2)^{3/2}}, \\ 6 \left(\frac{d^2}{dt^2} r_2(t) \right) &= -\frac{6 (r_2(t) - q_2(t))}{((r_1(t) - q_1(t))^2 + (r_2(t) - q_2(t))^2)^{3/2}}, \frac{d^2}{dt^2} q_1(t) = \\ &-\frac{6 (q_1(t) - r_1(t))}{((q_1(t) - r_1(t))^2 + (q_2(t) - r_2(t))^2)^{3/2}}, \frac{d^2}{dt^2} q_2(t) = \\ &-\frac{6 (q_2(t) - r_2(t))}{((q_1(t) - r_1(t))^2 + (q_2(t) - r_2(t))^2)^{3/2}} \end{aligned} \right\} \quad (2.3)$$

Define initial conditions (initial positions and velocities for both bodies):

```
> qrincon :=
[[q[1](0)=5, q[2](0)=0, D(q[1])(0) = 0.2, D(q[2])(0) = 1,
r[1](0)=0, r[2](0)=1, D(r[1])(0) = 0.2, D(r[2])(0) = -1/6]
];
```

$$qrincon := \left[\left[q_1(0) = 5, q_2(0) = 0, D(q_1)(0) = 0.2, D(q_2)(0) = 1, r_1(0) = 0, r_2(0) = 1, \right. \right. \quad (2.4)$$

$$\left. \left. D(r_1)(0) = 0.2, D(r_2)(0) = -\frac{1}{6} \right] \right]$$


The system of ODE is complicated to solve, hence we apply Maple's numerical solver:

```
> qnum := DEplot(grav_ode, [q[1],q[2],r[1],r[2]], t=0..100,
qrincon,
                stepsize=0.2, scene=[q[1],q[2]], linecolor=
blue):
rnum := DEplot(grav_ode, [q[1],q[2],r[1],r[2]], t=0..100,
```

```

qrincon,
                stepsize=0.2, scene=[r[1],r[2]], linecolor=
red):
display(qnum,rnum);

```



▼ Solution in a Local Coordinate System

We try to plot how the solution looks like as observed from one of the bodies (for example, the movement of the Moon, as seen from the Earth)

For that purpose, we need dsolve to generate the numerical solution as a procedure:

```

> qrsol := dsolve( { op(grav_ode), op(qrincon[1]) },

```

```

numeric,
      {q[1](t), q[2](t), r[1](t), r[2](t)}, output=
listprocedure);
qrsol := [ t = proc(t) ... end proc, q1(t) = proc(t) ... end proc,  $\frac{d}{dt} q_1(t) = \text{proc}(t)$  (2.1.1)
...
end proc, q2(t) = proc(t) ... end proc,  $\frac{d}{dt} q_2(t) = \text{proc}(t)$  ... end proc, r1(t) =
proc(t)
...
end proc,  $\frac{d}{dt} r_1(t) = \text{proc}(t)$  ... end proc, r2(t) = proc(t) ... end proc,  $\frac{d}{dt} r_2(t) =$ 
proc(t)
...
end proc ]


```

From these four procedures generated by Maple (two procedures each for positions and velocities of the bodies), we only need the position functions:

```

> qs1 := subs(qrsol, q[1](t)): qs2 := subs(qrsol, q[2](t)):
rs1 := subs(qrsol, r[1](t)): rs2 := subs(qrsol, r[2](t)):
> qplot := plot( [ qs1,qs2, 0..100], scaling=CONSTRAINED,
colour=blue, thickness=2 ):
rplot := plot( [ rs1,rs2, 0..100], scaling=CONSTRAINED,
color=red, thickness=2 ):
display(qplot,rplot);

```



The motion of one of the bodies relative to the other one is then obtained as the difference of the two functions

(note that the result is not simply a circular motion around the central body):

```
> ds1 := qs1-rs1; ds2 := qs2-rs2;
  plot( [ ds1,ds2, 0..100], scaling=CONSTRAINED);
      ds1 := qs1 - rs1
      ds2 := qs2 - rs2
```